

## NAG Library Routine Document

### F08QLF (DTRSNA)

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

## 1 Purpose

F08QLF (DTRSNA) estimates condition numbers for specified eigenvalues and/or right eigenvectors of a real upper quasi-triangular matrix.

## 2 Specification

```

SUBROUTINE F08QLF (JOB, HOWMNY, SELECT, N, T, LDT, VL, LDVL, VR, LDVR,      &
                  S, SEP, MM, M, WORK, LDWORK, IWORK, INFO)
INTEGER           N, LDT, LDVL, LDVR, MM, M, LDWORK, IWORK(*), INFO
REAL (KIND=nag_wp) T(LDT,*), VL(LDVL,*), VR(LDVR,*), S(*), SEP(*),      &
                  WORK(LDWORK,*)
LOGICAL          SELECT(*)
CHARACTER(1)     JOB, HOWMNY

```

The routine may be called by its LAPACK name *dtrsna*.

## 3 Description

F08QLF (DTRSNA) estimates condition numbers for specified eigenvalues and/or right eigenvectors of a real upper quasi-triangular matrix  $T$  in canonical Schur form. These are the same as the condition numbers of the eigenvalues and right eigenvectors of an original matrix  $A = ZTZ^T$  (with orthogonal  $Z$ ), from which  $T$  may have been derived.

F08QLF (DTRSNA) computes the reciprocal of the condition number of an eigenvalue  $\lambda_i$  as

$$s_i = \frac{|v^H u|}{\|u\|_E \|v\|_E},$$

where  $u$  and  $v$  are the right and left eigenvectors of  $T$ , respectively, corresponding to  $\lambda_i$ . This reciprocal condition number always lies between zero (i.e., ill-conditioned) and one (i.e., well-conditioned).

An approximate error estimate for a computed eigenvalue  $\lambda_i$  is then given by

$$\frac{\epsilon \|T\|}{s_i},$$

where  $\epsilon$  is the *machine precision*.

To estimate the reciprocal of the condition number of the right eigenvector corresponding to  $\lambda_i$ , the routine first calls F08QFF (DTREXC) to reorder the eigenvalues so that  $\lambda_i$  is in the leading position:

$$T = Q \begin{pmatrix} \lambda_i & c^T \\ 0 & T_{22} \end{pmatrix} Q^T.$$

The reciprocal condition number of the eigenvector is then estimated as  $sep_i$ , the smallest singular value of the matrix  $(T_{22} - \lambda_i I)$ . This number ranges from zero (i.e., ill-conditioned) to very large (i.e., well-conditioned).

An approximate error estimate for a computed right eigenvector  $u$  corresponding to  $\lambda_i$  is then given by

$$\frac{\epsilon \|T\|}{sep_i}.$$

## 4 References

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

## 5 Arguments

- 1: JOB – CHARACTER(1) *Input*  
*On entry:* indicates whether condition numbers are required for eigenvalues and/or eigenvectors.  
 JOB = 'E'  
     Condition numbers for eigenvalues only are computed.  
 JOB = 'V'  
     Condition numbers for eigenvectors only are computed.  
 JOB = 'B'  
     Condition numbers for both eigenvalues and eigenvectors are computed.  
*Constraint:* JOB = 'E', 'V' or 'B'.
- 2: HOWMNY – CHARACTER(1) *Input*  
*On entry:* indicates how many condition numbers are to be computed.  
 HOWMNY = 'A'  
     Condition numbers for all eigenpairs are computed.  
 HOWMNY = 'S'  
     Condition numbers for selected eigenpairs (as specified by SELECT) are computed.  
*Constraint:* HOWMNY = 'A' or 'S'.
- 3: SELECT(\*) – LOGICAL array *Input*  
**Note:** the dimension of the array SELECT must be at least  $\max(1, N)$  if HOWMNY = 'S', and at least 1 otherwise.  
*On entry:* specifies the eigenpairs for which condition numbers are to be computed if HOWMNY = 'S'. To select condition numbers for the eigenpair corresponding to the real eigenvalue  $\lambda_j$ , SELECT(*j*) must be set .TRUE.. To select condition numbers corresponding to a complex conjugate pair of eigenvalues  $\lambda_j$  and  $\lambda_{j+1}$ , SELECT(*j*) and/or SELECT(*j* + 1) must be set to .TRUE..  
 If HOWMNY = 'A', SELECT is not referenced.
- 4: N – INTEGER *Input*  
*On entry:* *n*, the order of the matrix *T*.  
*Constraint:*  $N \geq 0$ .
- 5: T(LDT, \*) – REAL (KIND=nag\_wp) array *Input*  
**Note:** the second dimension of the array T must be at least  $\max(1, N)$ .  
*On entry:* the *n* by *n* upper quasi-triangular matrix *T* in canonical Schur form, as returned by F08PEF (DHSEQR).
- 6: LDT – INTEGER *Input*  
*On entry:* the first dimension of the array T as declared in the (sub)program from which F08QLF (DTRSNA) is called.  
*Constraint:*  $LDT \geq \max(1, N)$ .

- 7: VL(LDVL,\*) – REAL (KIND=nag\_wp) array Input  
**Note:** the second dimension of the array VL must be at least  $\max(1, MM)$  if JOB = 'E' or 'B' and at least 1 if JOB = 'V'.  
*On entry:* if JOB = 'E' or 'B', VL must contain the left eigenvectors of  $T$  (or of any matrix  $QTQ^T$  with  $Q$  orthogonal) corresponding to the eigenpairs specified by HOWMNY and SELECT. The eigenvectors **must** be stored in consecutive columns of VL, as returned by F08PKF (DHSEIN) or F08QKF (DTREVC).  
 If JOB = 'V', VL is not referenced.
- 8: LDVL – INTEGER Input  
*On entry:* the first dimension of the array VL as declared in the (sub)program from which F08QLF (DTRSNA) is called.  
*Constraints:*  
     if JOB = 'E' or 'B',  $LDVL \geq \max(1, N)$ ;  
     if JOB = 'V',  $LDVL \geq 1$ .
- 9: VR(LDVR,\*) – REAL (KIND=nag\_wp) array Input  
**Note:** the second dimension of the array VR must be at least  $\max(1, MM)$  if JOB = 'E' or 'B' and at least 1 if JOB = 'V'.  
*On entry:* if JOB = 'E' or 'B', VR must contain the right eigenvectors of  $T$  (or of any matrix  $QTQ^T$  with  $Q$  orthogonal) corresponding to the eigenpairs specified by HOWMNY and SELECT. The eigenvectors **must** be stored in consecutive columns of VR, as returned by F08PKF (DHSEIN) or F08QKF (DTREVC).  
 If JOB = 'V', VR is not referenced.
- 10: LDVR – INTEGER Input  
*On entry:* the first dimension of the array VR as declared in the (sub)program from which F08QLF (DTRSNA) is called.  
*Constraints:*  
     if JOB = 'E' or 'B',  $LDVR \geq \max(1, N)$ ;  
     if JOB = 'V',  $LDVR \geq 1$ .
- 11: S(\*) – REAL (KIND=nag\_wp) array Output  
**Note:** the dimension of the array S must be at least  $\max(1, MM)$  if JOB = 'E' or 'B', and at least 1 otherwise.  
*On exit:* the reciprocal condition numbers of the selected eigenvalues if JOB = 'E' or 'B', stored in consecutive elements of the array. Thus  $S(j)$ ,  $SEP(j)$  and the  $j$ th columns of VL and VR all correspond to the same eigenpair (but not in general the  $j$ th eigenpair unless all eigenpairs have been selected). For a complex conjugate pair of eigenvalues, two consecutive elements of S are set to the same value.  
 If JOB = 'V', S is not referenced.
- 12: SEP(\*) – REAL (KIND=nag\_wp) array Output  
**Note:** the dimension of the array SEP must be at least  $\max(1, MM)$  if JOB = 'V' or 'B', and at least 1 otherwise.  
*On exit:* the estimated reciprocal condition numbers of the selected right eigenvectors if JOB = 'V' or 'B', stored in consecutive elements of the array. For a complex eigenvector, two consecutive elements of SEP are set to the same value. If the eigenvalues cannot be reordered to compute  $SEP(j)$ , then  $SEP(j)$  is set to zero; this can only occur when the true value would be very small anyway.

If JOB = 'E', SEP is not referenced.

- 13: MM – INTEGER *Input*  
*On entry:* the number of elements in the arrays S and SEP, and the number of columns in the arrays VL and VR (if used). The precise number required,  $m$ , is  $n$  if HOWMNY = 'A'; if HOWMNY = 'S',  $m$  is obtained by counting 1 for each selected real eigenvalue, and 2 for each selected complex conjugate pair of eigenvalues (see SELECT), in which case  $0 \leq m \leq n$ .  
*Constraint:*  $MM \geq M$ .
- 14: M – INTEGER *Output*  
*On exit:*  $m$ , the number of elements of S and/or SEP actually used to store the estimated condition numbers. If HOWMNY = 'A', M is set to  $n$ .
- 15: WORK(LDWORK,\*) – REAL (KIND=nag\_wp) array *Workspace*  
**Note:** the second dimension of the array WORK must be at least  $\max(1, N + 6)$  if JOB = 'V' or 'B' and at least 1 if JOB = 'E'.  
 If JOB = 'E', WORK is not referenced.
- 16: LDWORK – INTEGER *Input*  
*On entry:* the first dimension of the array WORK as declared in the (sub)program from which F08QLF (DTRSNA) is called.  
*Constraints:*  
     if JOB = 'V' or 'B',  $LDWORK \geq \max(1, N)$ ;  
     if JOB = 'E',  $LDWORK \geq 1$ .
- 17: IWORK(\*) – INTEGER array *Workspace*  
**Note:** the dimension of the array IWORK must be at least  $\max(1, 2 \times (N - 1))$ .
- 18: INFO – INTEGER *Output*  
*On exit:* INFO = 0 unless the routine detects an error (see Section 6).

## 6 Error Indicators and Warnings

INFO < 0

If INFO =  $-i$ , argument  $i$  had an illegal value. An explanatory message is output, and execution of the program is terminated.

## 7 Accuracy

The computed values  $sep_i$  may over estimate the true value, but seldom by a factor of more than 3.

## 8 Parallelism and Performance

F08QLF (DTRSNA) makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

## 9 Further Comments

For a description of canonical Schur form, see the document for F08PEF (DHSEQR).

The complex analogue of this routine is F08QYF (ZTRSNA).

## 10 Example

This example computes approximate error estimates for all the eigenvalues and right eigenvectors of the matrix  $T$ , where

$$T = \begin{pmatrix} 0.7995 & -0.1144 & 0.0060 & 0.0336 \\ 0.0000 & -0.0994 & 0.2478 & 0.3474 \\ 0.0000 & -0.6483 & -0.0994 & 0.2026 \\ 0.0000 & 0.0000 & 0.0000 & -0.1007 \end{pmatrix}.$$

### 10.1 Program Text

```

Program f08qlfe

!      F08QLF Example Program Text

!      Mark 26 Release. NAG Copyright 2016.

!      .. Use Statements ..
Use nag_library, Only: dlange => f06raf, dtrevc, dtrsna, nag_wp, x02ajf
!      .. Implicit None Statement ..
Implicit None
!      .. Parameters ..
Integer, Parameter          :: nin = 5, nout = 6
!      .. Local Scalars ..
Real (Kind=nag_wp)         :: eps, tnorm
Integer                     :: i, info, ldt, ldvl, ldvr, ldwork, m, &
                             n
!      .. Local Arrays ..
Real (Kind=nag_wp), Allocatable :: s(:), sep(:), t(:,,:), vl(:,,:),      &
                             vr(:,,:), work(:,,:)
Integer, Allocatable         :: iwork(:)
Logical                     :: select(1)
!      .. Executable Statements ..
Write (nout,*) 'F08QLF Example Program Results'
Write (nout,*)
!      Skip heading in data file
Read (nin,*)
Read (nin,*) n
ldt = n
ldvl = n
ldvr = n
ldwork = n
Allocate (s(n),sep(n),t(ldt,n),vl(ldvl,n),vr(ldvr,n),work(ldwork,n+6), &
         iwork(2*n-1))

!      Read T from data file

Read (nin,*)(t(i,1:n),i=1,n)

!      Calculate the left and right eigenvectors of T

!      The NAG name equivalent of dtrevc is f08qkf
Call dtrevc('Both','All',select,n,t,ldt,vl,ldvl,vr,ldvr,n,m,work,info)

!      Estimate condition numbers for all the eigenvalues and right
!      eigenvectors of T

!      The NAG name equivalent of dtrsna is f08qlf
Call dtrsna('Both','All',select,n,t,ldt,vl,ldvl,vr,ldvr,s,sep,n,m,work, &
         ldwork,iwork,info)

```

```

!      Print condition numbers of eigenvalues and right eigenvectors

      Write (nout,*) 'S'
      Write (nout,99999) s(1:m)
      Write (nout,*)
      Write (nout,*) 'SEP'
      Write (nout,99999) sep(1:m)

!      Calculate approximate error estimates (using the 1-norm)

      eps = x02ajf()
!      f06raf is the NAG name equivalent of the LAPACK auxiliary dlange
      tnorm = dlange('1-norm',n,n,t,ldt,work)
      Write (nout,*)
      Write (nout,*) 'Approximate error estimates for eigenvalues ',      &
        'of T (machine-dependent)'
      Write (nout,99999)(eps*tnorm/s(i),i=1,m)
      Write (nout,*)
      Write (nout,*) 'Approximate error estimates for right ',      &
        'eigenvectors of T (machine-dependent)'
      Write (nout,99999)(eps*tnorm/sep(i),i=1,m)

99999 Format ((3X,1P,7E11.1))
      End Program f08qlfe

```

## 10.2 Program Data

```

F08QLF Example Program Data
4                                     :Value of N
0.7995  -0.1144  0.0060  0.0336
0.0000  -0.0994  0.2478  0.3474
0.0000  -0.6483  -0.0994  0.2026
0.0000  0.0000  0.0000  -0.1007      :End of matrix T

```

## 10.3 Program Results

```

F08QLF Example Program Results

S
    9.9E-01    7.0E-01    7.0E-01    5.7E-01

SEP
    6.3E-01    3.7E-01    3.7E-01    3.1E-01

Approximate error estimates for eigenvalues of T (machine-dependent)
    9.6E-17    1.4E-16    1.4E-16    1.7E-16

Approximate error estimates for right eigenvectors of T (machine-dependent)
    1.5E-16    2.6E-16    2.6E-16    3.1E-16

```

---